Riemann Sums

Objective

This lab emphasizes the graphical and numerical aspects of Riemann sums. Visually, it will be apparent that Riemann sums converge to the "area" under the graph of a function. Numerically, the values of the Riemann sums converge.

Background

The `ApproximateIntegration` maplet provides numerical values and visual pictures — including animations — for more general Riemann sums.

There are three Maple commands for working with sums. The `add` command is used to add a finite and explicit sequence of expressions. The `sum` command is designed for the evaluation of symbolic sums (i.e., sums with indefinite, including infinite, limits of summation). The `Sum` command is the *inert* version of `sum`. An inert command is not evaluated or otherwise manipulated until explicitly commanded to do so — typically with `value` or `evalf`. Other inert commands are `Limit`, `Diff`, and `Int`. Maple's `collect` command is used to rewrite an expression in terms of powers of the indicated variables.

Riemann sums are, however, not the primary tool used to evaluate definite integral. Example 2 and Question 3 provide evidence that there is a more general and functional connection between integration and differentiation. The Fundamental Theorems of Calculus are the missing link(s).

Discussion

Example 1: Using the `ApproximateIntegration` Maplet

- launch the `ApproximateIntegration` maplet
- in the **Function** box, enter `x^2 + 1`
- in the **a =** and **b =** boxes, enter `-1` and `2`, respectively
- in the **Riemann Sums** region, check that `midpoint` is selected
- press the **Plot** button
- slide the **Number of Partition** slider and press the **Plot** button until you have a partition with 6 subintervals
- change the sample points from `midpoint` to `left` and press the **Plot** button
- change the sample points from `left` to `right` and press the **Plot** button
- change the sample points from `right` to `random` and press the **Plot** button
- slide the **Number of Partition** slider and press the **Plot** button until you have a partition with 3 subintervals
- in the **Number of Frames** box, enter `5`
- check that the **Repeat Animation** box is checked, the **Subdivide Interval** setting is **random**, and the **Subpartition** setting is **all**, then press the **Animate** button

Enter, and execute, the following Maple commands in a Maple worksheet.

Example 2: Evaluation of $\int_a^b x^2\,dx$ via Riemann Sums

```
> g := x -> x^2;                              # define integrand
> Delta[x] := (b-a)/N;                        # norm of partition, |P|
> x[i] := a + i*Delta[x];                     # sample pts for right R sum
> q6 := collect( g( x[i] ), i );              # summand (grouped by index, i)
> q7 := Delta[x] * Sum( q6, i=1..N );         # right R sum with N subint
> q8 := value( q7 );                          # value of right R sum
> q9 := collect( q8, N );                     # regroup by N
> q10 := limit( q9, N=infinity );             # limit of R sum as |P| → 0
> collect( q10, {a,b} );                      # simplified value of ∫_a^b x^2 dx
```

Example 3: The `add`, `sum`, and `Sum` Commands

```
> f := k -> 1/k^2;                          # define summand
> q1 := add( f(i), i=1..4 ):               # 4-term sum
> [ q1, value(q1), evalf(q1) ];             # evaluations of unevaluated sum
> q2 := sum( f(i), i=1..4 ):               # evaluated 4-term sum
> [ q2, value(q2), evalf(q2) ];             # evaluations of evaluated sum
> q3 := Sum( f(i), i=1..4 ):               # unevaluated 4-term sum
> [ q3, value(q3), evalf(q3) ];             # evaluations of explicit sum
> q4 := Sum( f(i), i=1..infinity );        # unevaluated infinite sum
> [ q4, value(q4), evalf(q4) ];             # evaluations of infinite sum
> q5 := add( f(i), i=1..infinity );        # error!
```

Notes

(1) The `ApproximateInt` command from the `Student[Calculus1]` package, produces the results for the `ApproximateIntegration` maplet.

(2) Animations created with the `ApproximateIntegration` maplet are controlled with the **Play**, **Stop**, and **Pause** buttons. Note that the value of each Riemann sum is included at the bottom of the plot region. The value of the first Riemann sum is shown in the **Area =** box.

(3) The **Compare** button on the `ApproximateIntegration` maplet window creates a new window in which the numerical value of five different Riemann sums are compared. The other values reported in this window are for more sophisticated numerical methods for approximating definite integrals. Some of these are discussed in Calculus II and Numerical Analysis.

(4) The concluding 4 steps in the evaluation of $\int_a^b x^2\,dx$ in Example 2 can be replaced with

```
> q8a := Limit( q7, N=infinity );            # limit of R sum as |P| -> 0
> q9a = collect( value( q8a ), {a,b} );      # simplified value of int_a^b x^2 dx
```

Questions

(1) (a) Create a table of numerical values of the Riemann sums for $\sqrt{x}$ on $[0,4]$ on partitions with 1, 2, 4, 8, 16, 32, 64, 128, and 256 random-width subintervals with randomly selected sample points.

(b) Use Riemann Sums to determine the exact value of $\displaystyle\int_0^4 \sqrt{x}\,dx$.

(2) Consider the definite integral $\displaystyle\int_1^5 \frac{1}{x}\,dx$.

(a) Find the values of the 10-subinterval Riemann sums using upper, lower, left, right, and midpoint sample points. Why do the left and upper Riemann sums agree? Why do the right and lower Riemann sums agree?

(b) Examine these five Riemann sums with more (many more!) subintervals. Which of these values is the best approximation to the exact value of the definite integral?

(3) Use Riemann Sums to determine explicit formulae for each of the following definite integrals:

$$\int_a^b x^6\,dx \qquad \int_a^b \sqrt{x}\,dx \qquad \int_a^b \frac{x}{\sqrt{x^2+1}}\,dx. \qquad \int_a^b \cos x\,dx$$